Appendix B Calculating Exceedances for a N-S Critical Load Function

In the case of a critical load function (as defined in Chapter 3) there is no unique exceedance. This is illustrated in Figure B1: Let the point E denote the deposition of N and S. By reducing N_{dep} substantially, one reaches the point Z_1 and thus non-exceedance without reducing S_{dep}; on the other hand one can reach non-exceedance by only reducing S_{dep} until reaching Z_3; finally, with a reduction of both N_{dep} and S_{dep}, one can reach non-exceedance as well (e.g. point Z_2).

Intuitively, the reduction required in N and S deposition to reach point Z_2 (see Figure B1), i.e. the shortest distance to the critical load function, seems a good measure for exceedance. Thus we *define* the exceedance for a given pair of depositions (N_{dep},S_{dep}) as the sum of the N and S deposition reductions required to reach the critical load function by the 'shortest' path. Figure B2 depicts the cases that can arise: if the deposition falls …

(a) … on or below the critical load function (Region 0). In this case the exceedance is defined as zero (non-exceedance);

(b) … into Region 1 (e.g. point E_1): An S deposition reduction does not help; an N deposition reduction is needed: the exceedance is defined as N_{dep}–CLN_{max};

(c) … into Region 2 (e.g. point E_2): the exceedance in this region is defined as the sum of N and S deposition reduction needed to reach the corner-point point Z_2;

(d) … into Region 3 (e.g. point E_3): the exceedance is given by the sum of N and S deposition reduction, *ExN*+*ExS*, required to reach the point Z_3, with the line E_3–Z_3 perpendicular to the CLF;

Figure B1. Critical load function (CLF) of N and S (thick line). The grey-shaded area below the critical load function defines deposition pairs (N_{dep},S_{dep}) for which there is non-exceedance. The points E and Z_1-Z_3 demonstrate that non-exceedance can be attained in different ways, i.e. there is no unique exceedance.



Figure B2. Illustration of the different cases for calculating the exceedance for a given critical load function.



(e)… into Region 4 (e.g. point E_4): the exceedance is defined as the sum of N and S deposition reduction needed to reach the corner-point Z_4;

(f) … into Region 5 (e.g. point E_5): an N deposition reduction does not help; an S deposition reduction is needed: the exceedance is defined as $S_{dep}-CLS_{max}$.

The exceedance function $Ex(N_{dep},S_{dep})$ can be described by the following equation (the coordinates of the point Z_3 are denoted by (N_o,S_o)):

(B-1)
$$Ex(N_{dep},S_{dep}) = \begin{cases} 0 & \text{if } (N_{dep},S_{dep}) \in \text{Region 0} \\ N_{dep}-CLN_{max} & \text{if } (N_{dep},S_{dep}) \in \text{Region 1} \\ N_{dep}-CLN_{max}+S_{dep}-CLS_{min} & \text{if } (N_{dep},S_{dep}) \in \text{Region 2} \\ N_{dep}-N_0+S_{dep}-S_0 & \text{if } (N_{dep},S_{dep}) \in \text{Region 3} \\ N_{dep}-CLN_{min}+S_{dep}-CLS_{max} & \text{if } (N_{dep},S_{dep}) \in \text{Region 4} \\ S_{dep}-CLS_{max} & \text{if } (N_{dep},S_{dep}) \in \text{Region 5} \end{cases}$$

The computation of the exceedance function requires the coordinates of the point (N_o,S_o) on the critical load function. If (x_1,y_1) and (x_2,y_2) are two arbitrary points of a straight line *g* and (x_e,y_e) another point (not on that line), then the coordinates $(x_o,y_o) = (N_o,S_o)$ of the point obtained by intersecting the line passing through (x_e,y_e) and perpendicular to *g* are given by:

(B-2a) $\quad x_0 = (d_1 s + d_2 v)/d^2 \quad \text{and} \quad y_0 = (d_2 s - d_1 v)/d^2$

with

(B-2b) $\quad d_1 = x_2 - x_1\,,\;\; d_2 = y_2 - y_1\,,\;\; d^2 = d_1^2 + d_2^2$

and

(B-2c) $s = x_e d_1 + y_e d_2\,,\;\; v = x_1 d_2 - y_1 d_1 = x_1 y_2 - y_1 x_2$

The final difficulty in computing the *Ex*(N_{dep},S_{dep}) is to determine into which of the regions (Region 0 through Region 5 in Figure B2) a given pair of deposition (N_{dep},S_{dep}) falls. Without going into the details of the geometrical considerations, a FORTRAN subroutine is listed below, which returns the number of the region as well as *ExN* and *ExS*:

```
subroutine  exceedNS  (CLNmin,CLSmax,CLNmax,CLSmin,depN,depS,ExN,ExS,ireg)
!
! Returns - in double precision - the exceedances ExN and ExS (Ex=ExN+ExS)
! for double-precision N and S depositions depN and depS and the CLF
! defined by (CLNmin,CLSmax) and (CLNmax,CLSmin).
! The "region" in which (depN,depS) lies, is returned in ireg.
!
  implicit none
!
  real,    intent(in)  :: CLNmin, CLSmax, CLNmax, CLSmin
  real(8), intent(in)  :: depN, depS
  real(8), intent(out) :: ExN, ExS
  integer, intent(out) :: ireg
!
  real(8)              :: dN, dS, dd, s, v, xf, yf
!
  ExN = -1; ExS = -1; ireg = -1
  if (CLNmin < 0 .or. CLSmax < 0 .or. CLNmax < 0 .or. CLSmin < 0) return
  ExN = depN; ExS = depS; ireg = 9
!  CLN = CLNmax
  if (CLSmax == 0 .and. CLNmax == 0)    return
!  CLS = CLSmin
  dN = CLNmin-CLNmax
  dS = CLSmax-CLSmin
  if (depS <= CLSmax .and. depN <= CLNmax .and. &
&     (depN-CLNmax)*dS <= (depS-CLSmin)*dN) then ! non-exceedance:
    ireg = 0
    ExN = 0; ExS = 0
  else if (depS <= CLSmin) then
    ireg = 1
    ExN = depN-CLNmax; ExS = 0
  else if (depN <= CLNmin) then
    ireg = 5
    ExN = 0; ExS = depS-CLSmax
  else if (-(depN-CLNmax)*dN >=(depS-CLSmin)*dS) then
    ireg = 2
    ExN = depN-CLNmax; ExS = depS-CLSmin
  else if (-(depN-CLNmin)*dN <= (depS-CLSmax)*dS) then
    ireg = 4
    ExN = depN-CLNmin; ExS = depS-CLSmax
  else
    ireg = 3
    dd = dN*dN+dS*dS
    s = depN*dN+depS*dS
    v = CLNmax*dS-CLSmin*dN
    xf = (dN*s+dS*v)/dd
    yf = (dS*s-dN*v)/dd
    ExN = depN-xf; ExS = depS-yf
  end if
                                          return
end subroutine exceedNS
```